Time varying system

Example 1

Let us introduce the following Ore algebra

```
A = OreAlgebra[Der[t]]
```

$\mathbb{K}(t)[D_t; 1, D_t]$

and the two matrices Emat and Fmat:

```
Emat = ToOrePolynomial[{{0, 1}, {-1, 0}}, A]
```

```
{{0, 1}, {-1, 0}}
```

```
Fmat = ToOrePolynomial[{{Cos[t]}, {Sin[t]}}, A]
```

```
{{Cos[t]}, {Sin[t]}}
```

We can form the matrix of differential operators R = (Der[t] I - Emat, -Fmat) which corresponds to the system $\dot{x}$ = Emat x + Fmat u.

```
MatrixForm[
 R = Join[ToOrePolynomial[DiagonalMatrix[{Der[t], Der[t]}], A] - Emat, -Fmat, 2]]
```

$$\begin{pmatrix} D_t & -1 & -\text{Cos}[t] \\ 1 & D_t & -\text{Sin}[t] \end{pmatrix}$$

Let us introduce the finitely presented left A-module M $= A^{1\times 3} / A^{1\times 2} R$.
Let us now compute the adjoint Radj of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t & 1 \\ -1 & -D_t \\ -\text{Cos}[t] & -\text{Sin}[t] \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} -1 & -D_t & \text{Sin}[t] \\ -D_t & 1 & \text{Cos}[t] \end{pmatrix}, \begin{pmatrix} -\text{Cos}[t]\, D_t - 2\,\text{Sin}[t] \\ -\text{Sin}[t]\, D_t + 2\,\text{Cos}[t] \\ -D_t^2 - 4 \end{pmatrix}\right\}$$

Since the first matrix is identity, M is torsion-free. Hence the corresponding system is controllable.
Let us now check whether or not M is free.

```
T = Simplify[LeftInverse[Q, A]]
```

$$\left\{\left\{-\frac{\text{Sin}[t]}{2}, \frac{\text{Cos}[t]}{2}, 0\right\}\right\}$$

Since the parametrization Q admits a left inverse, we know that M is a free left A-module of rank 1, i.e. the corresponding system is flat. Finally a flat output ξ is defined by

```
ApplyMatrix[T, {x1[t], x2[t], u[t]}][[1]]
```

$\frac{1}{2}(-\mathrm{Sin}[t]\, x_1[t] + \mathrm{Cos}[t]\, x_2[t])$

and an injective parametrization of the flat system is given by:

```
Thread[{x1[t], x2[t], u[t]} -> ApplyMatrix[Q, {ξ[t]}]]
```

$\{x_1[t] \to -2\,\mathrm{Sin}[t]\,\xi[t] - \mathrm{Cos}[t]\,\xi'[t],$
$x_2[t] \to 2\,\mathrm{Cos}[t]\,\xi[t] - \mathrm{Sin}[t]\,\xi'[t],\ u[t] \to -4\,\xi[t] - \xi''[t]\}$

Example 2

Let us introduce the following Ore algebra

```
A = OreAlgebra[Der[t]]
```

$\mathbb{K}(t)[D_t; 1, D_t]$

and the two matrices Emat and Fmat:

```
Emat = ToOrePolynomial[{{0, 1}, {-1, 0}}, A]
```

{{0, 1}, {-1, 0}}

```
Fmat = ToOrePolynomial[{{Cos[t]}, {-Sin[t]}}, A]
```

{{Cos[t]}, {-Sin[t]}}

We can form the matrix of differential operators R = (Der[t] I - Emat, -Fmat) which corresponds to the system $\dot{x}$ = Emat x + Fmat u.

```
MatrixForm[
 R = Join[ToOrePolynomial[DiagonalMatrix[{Der[t], Der[t]}], A] - Emat, -Fmat, 2]]
```

$$\begin{pmatrix} D_t & -1 & -\mathrm{Cos}[t] \\ 1 & D_t & \mathrm{Sin}[t] \end{pmatrix}$$

Let us introduce the finitely presented left A-module M $= A^{1\times 3} / A^{1\times 2} R$.
Let us now compute the adjoint Radj of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t & 1 \\ -1 & -D_t \\ -\mathrm{Cos}[t] & \mathrm{Sin}[t] \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{\begin{pmatrix} D_t & 0 \\ 0 & D_t \end{pmatrix}, \begin{pmatrix} \mathrm{Sin}[t] & \mathrm{Cos}[t] & 0 \\ 0 & -\mathrm{Sin}[t]\,D_t + \mathrm{Cos}[t] & -\mathrm{Sin}[t]^2 \end{pmatrix}, \begin{pmatrix} \mathrm{Cos}[t] \\ -\mathrm{Sin}[t] \\ D_t \end{pmatrix}\right\}$$

Since the first matrix is not identity, we deduce that M admits nontrivial torsion elements and thus the corresponding system admits autonomous elements $\tau[1]$ and $\tau[2]$, defined by:

```
AutonomousElements[R, {x1[t], x2[t], u[t]}, τ, A, Relations → True]
```

$\{\{\tau[1][t] \to \mathrm{Sin}[t]\,x_1[t] + \mathrm{Cos}[t]\,x_2[t],$
$\tau[2][t] \to -\mathrm{Sin}[t]^2\,u[t] + \mathrm{Cos}[t]\,x_2[t] - \mathrm{Sin}[t]\,x_2'[t]\},\ \{\tau[1]'[t] == 0,\ \tau[2]'[t] == 0\},$
$\{\mathrm{Csc}[t]\,(-\mathrm{Cot}[t]\,\tau[1][t] + \mathrm{Cot}[t]\,\tau[2][t] + \tau[1]'[t]),\ \mathrm{Csc}[t]\,(\tau[1][t] - \tau[2][t])\}\}$

The first list gives the definition of the autonomous elements τ[1] and τ[2], the second list gives the equations they satisfy and the last list gives the relations between τ[1] and τ[2].

The controllable part of the system is defined by:

```
Thread[ApplyMatrix[Rp, {x1[t], x2[t], u[t]}] == 0]
```

$\{\mathrm{Sin}[t]\, x_1[t] + \mathrm{Cos}[t]\, x_2[t] == 0,\ -\mathrm{Sin}[t]^2\, u[t] + \mathrm{Cos}[t]\, x_2[t] - \mathrm{Sin}[t]\, x_2'[t] == 0\}$

Finally let us compute a flat output of the controllable part:

```
ApplyMatrix[LeftInverse[Q, A], {x1[t], x2[t], u[t]}][[1]]
```

$-\mathrm{Csc}[t]\, x_2[t]$

Example 3

Let us introduce the following Ore algebra

```
A = OreAlgebra[Der[t]]
```

$\mathbb{K}(t)[D_t; 1, D_t]$

and the two matrices Emat and Fmat:

```
Emat = ToOrePolynomial[{{t, 1, 0}, {0, t^3, 0}, {0, 0, t^2}}, A]
```

$\{\{t, 1, 0\}, \{0, t^3, 0\}, \{0, 0, t^2\}\}$

```
Fmat = ToOrePolynomial[{{0}, {1}, {1}}, A]
```

{{0}, {1}, {1}}

We can form the matrix of differential operators R = (Der[t] I - Emat, -Fmat) which corresponds to the system $\dot{x}$ = Emat x + Fmat u.

```
MatrixForm[R = Join[
   ToOrePolynomial[DiagonalMatrix[{Der[t], Der[t], Der[t]}], A] - Emat, -Fmat, 2]]
```

$$\begin{pmatrix} D_t - t & -1 & 0 & 0 \\ 0 & D_t - t^3 & 0 & -1 \\ 0 & 0 & D_t - t^2 & -1 \end{pmatrix}$$

Let us introduce the finitely presented left A-module M $= A^{1\times 4} / A^{1\times 3} R$.

Let us now compute the adjoint Radj of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t - t & 0 & 0 \\ -1 & -D_t - t^3 & 0 \\ 0 & 0 & -D_t - t^2 \\ 0 & -1 & -1 \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
{Ann, Rp, Q} = Exti[Radj, A, 1];
MatrixForm[Ann]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

MatrixForm[Rp]

$$\begin{pmatrix} 0 & 0 & -D_t + t^2 & 1 \\ 0 & -D_t + t^3 & 0 & 1 \\ -D_t + t & 1 & 0 & 0 \end{pmatrix}$$

Q

$$\{\{(2 t - 3 t^2 + t^3 - 2 t^4 + t^5) D_t + (4 - 12 t + 6 t^2 - 18 t^3 + 13 t^4 - t^5 + 2 t^6 - t^7)\},$$
$$\{(2 t - 3 t^2 + t^3 - 2 t^4 + t^5) D_t^2 + (6 - 18 t + 7 t^2 - 23 t^3 + 17 t^4 + t^5 + t^6 - t^7) D_t + (-12 + 8 t - 42 t^2 + 46 t^3 + 13 t^4 - t^5 - 6 t^6 - 2 t^7 + t^8)\},$$
$$\{(2 t - 3 t^2 + t^3 - 2 t^4 + t^5) D_t^2 + (6 - 18 t + 7 t^2 - 21 t^3 + 12 t^4 + 5 t^5 - 2 t^6 + 2 t^7 - t^8) D_t + (-12 + 8 t - 42 t^2 + 42 t^3 + 18 t^4 - 11 t^5 + 8 t^6 - 6 t^7 - 2 t^8 + t^9)\},$$
$$\{(2 t - 3 t^2 + t^3 - 2 t^4 + t^5) D_t^3 + (8 - 24 t + 10 t^2 - 31 t^3 + 20 t^4 + 4 t^5 + t^7 - t^8) D_t^2 + (-30 + 22 t - 111 t^2 + 108 t^3 + 36 t^4 - 2 t^5 + 10 t^6 - 19 t^7 - t^9 + t^{10}) D_t + (8 - 84 t + 138 t^2 + 64 t^3 - 13 t^4 + 6 t^5 - 60 t^6 - 5 t^7 + t^8 + 6 t^9 + 2 t^{10} - t^{11})\}\}$$

Since the first matrix is identity, M is torsion-free. Hence the corresponding system is controllable. Let us now check whether or not M is free.

T = Simplify[LeftInverse[Q, A]]

$$\left\{\left\{\frac{1 - t}{(2 - 3 t + t^2 - 2 t^3 + t^4)^2}, \frac{1}{t^2 (2 - 3 t + t^2 - 2 t^3 + t^4)^2}, -\frac{1}{t^2 (2 - 3 t + t^2 - 2 t^3 + t^4)^2}, 0\right\}\right\}$$

Since the parametrization Q admits a left inverse, we know that M is a free left A-module of rank 1, i.e. the corresponding system is flat. Finally a flat output ξ is defined by

ApplyMatrix[T, {x₁[t], x₂[t], x₃[t], u[t]}][[1]]

$$\frac{-(-1 + t) t^2 x_1[t] + x_2[t] - x_3[t]}{t^2 (2 - 3 t + t^2 - 2 t^3 + t^4)^2}$$

and an injective parametrization of the flat system is given by:

Thread[{x₁[t], x₂[t], x₃[t], u[t]} -> ApplyMatrix[Q, {ξ[t]}]]

$$\{x_1[t] \to -(-4 + 12 t - 6 t^2 + 18 t^3 - 13 t^4 + t^5 - 2 t^6 + t^7)\, \xi[t] + t (2 - 3 t + t^2 - 2 t^3 + t^4)\, \xi'[t],$$
$$x_2[t] \to (-12 + 8 t - 42 t^2 + 46 t^3 + 13 t^4 - t^5 - 6 t^6 - 2 t^7 + t^8)\, \xi[t] + (6 - 18 t + 7 t^2 - 23 t^3 + 17 t^4 + t^5 + t^6 - t^7)\, \xi'[t] + t (2 - 3 t + t^2 - 2 t^3 + t^4)\, \xi''[t],$$
$$x_3[t] \to (-12 + 8 t - 42 t^2 + 42 t^3 + 18 t^4 - 11 t^5 + 8 t^6 - 6 t^7 - 2 t^8 + t^9)\, \xi[t] + (6 - 18 t + 7 t^2 - 21 t^3 + 12 t^4 + 5 t^5 - 2 t^6 + 2 t^7 - t^8)\, \xi'[t] + t (2 - 3 t + t^2 - 2 t^3 + t^4)\, \xi''[t],$$
$$u[t] \to (8 - 84 t + 138 t^2 + 64 t^3 - 13 t^4 + 6 t^5 - 60 t^6 - 5 t^7 + t^8 + 6 t^9 + 2 t^{10} - t^{11})\, \xi[t] + (-30 + 22 t - 111 t^2 + 108 t^3 + 36 t^4 - 2 t^5 + 10 t^6 - 19 t^7 - t^9 + t^{10})\, \xi'[t] + (8 - 24 t + 10 t^2 - 31 t^3 + 20 t^4 + 4 t^5 + t^7 - t^8)\, \xi''[t] + t (2 - 3 t + t^2 - 2 t^3 + t^4)\, \xi^{(3)}[t]\}$$

Finally, we note that the flat output is defined when $t^2 (2 - 3 t + t^2 - 2 t^3 + t^4)^2$ is not equal to zero, that is to say, when $t \neq 0$ and not equal to:

sol = Solve[2 - 3 t + t^2 - 2 t^3 + t^4 == 0, t]

$$\left\{\{t \to 2\},\ \left\{t \to -\left(\frac{2}{3\left(9+\sqrt{93}\right)}\right)^{1/3} + \frac{\left(\frac{1}{2}\left(9+\sqrt{93}\right)\right)^{1/3}}{3^{2/3}}\right\},\right.$$

$$\left\{t \to -\frac{\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(9+\sqrt{93}\right)\right)^{1/3}}{2\times 3^{2/3}} + \frac{1-i\sqrt{3}}{2^{2/3}\left(3\left(9+\sqrt{93}\right)\right)^{1/3}}\right\},$$

$$\left.\left\{t \to -\frac{\left(1-i\sqrt{3}\right)\left(\frac{1}{2}\left(9+\sqrt{93}\right)\right)^{1/3}}{2\times 3^{2/3}} + \frac{1+i\sqrt{3}}{2^{2/3}\left(3\left(9+\sqrt{93}\right)\right)^{1/3}}\right\}\right\}$$

N[sol]

$\{\{t \to 2.\}, \{t \to 0.682328\}, \{t \to -0.341164 - 1.16154\, i\}, \{t \to -0.341164 + 1.16154\, i\}\}$

Two last solutions should be excluded since they are complex numbers.